
Introduction 
The Vector dialect in MLIR provides a low-level but still machine-agnostic dialect that             
can be targeted by higher-level dialects as a first step to bridge the semantic gap               
between vector operations and their concrete implementation at the hardware ISA. In            
previous case-studies (initial and follow-up), we conducted an explorative analysis of           
the AVX512 instructions that are generated for most operations of the Vector dialect.  
 
Since then, a few memory-related operations have been added to the Vector dialect             
(gather/scatter, expandload/compressstore, maskedload/maskedstore).   
Most CPUs, GPUs, and TPUs provide specialized instructions for such memory           
operations, and it was important for the Vector dialect to support all such operations in a                
sufficiently machine-agnostic way. This document supplements the previous        
case-studies by exploring AVX512 code generation for the gather and scatter operation            
in the context of a sparse matrix times dense vector operation. 

Sparse Matrix Preliminaries 
A matrix with many zero elements is called a sparse matrix. In contrast, a matrix               
without this property is called a dense matrix. Both the storage requirements and             
computational time of operations on sparse matrices can be reduced substantially by            
taking advantage of the zeros (viz. only store nonzero elements, avoid redundant            
operations on zeros, and even take advantage of particular nonzero structures).           
Although a vast amount of literature on sparse computations exist, here we simply refer              
to the downloadable “A.J.C. Bik. Compiler Support for Sparse Matrix Computations.           
PhD thesis, Leiden University, 1996”, which contains an elaborate literature review,           
including an overview of common sparse storage schemes. 

Sparse Storage Schemes 
Two common sparse storage schemes are the general sparse row-wise scheme and            
the general column-wise scheme (also called CRS/CCS). These schemes store only           
the nonzero elements of a sparse matrix along the rows/columns consecutively in a             
one-dimensional floating-point value array together with the column/row index in a           
parallel one-dimensional integer index array. Two additional one-dimensional integer         
“pointer” arrays are used to find the starting and ending positions of each row/column.              
Since the “pointer” arrays control the row/column access, the rows/columns may appear            
in arbitrary order in the value and parallel index arrays, with even gaps between the               
rows/columns. Within a row/column, elements are stored consecutively, but may appear           
in arbitrary column/row index order.   1

 
For example, the following “sparse” 5x5 matrix 
 

    | a11 0   a13 a14 0   | 

    | 0   a22 0   0   a25 | 

    | a31 0   a33 a34 0   | 

    | a41 0   0   a44 0   | 

    |  0  0   0   0   a55 | 

 

is stored as follows in the sparse row-wise scheme (with value array AVAL, index array               
AIDX, and “pointer” arrays ALOW and AHGH). A sparse column-wise scheme would be             
similar, indexing the “pointer” arrays by columns instead. 

1 Many variants of these sparse storage schemes exist. Some use a pointer and length array, some use                  
only one pointer array by making subsequent pointers one more than the previous pointer (which places                
additional ordering constraints). Some allow for insertion and deletion. And so on. 

https://theses.liacs.nl/pdf/bik-96.pdf


 

        1   2   3 |   4   5 |   6   7   8 |   9  10 |  11  

----+-------------+---------+-------------+---------+-----+ 

AVAL| a11 a13 a14 | a22 a25 | a31 a33 a34 | a41 a44 | a55 |  

AIDX|  1    3   4 |   2   5 |   1   3   4 |   1   4 |   5 | 

----+-------------+---------+-------------+---------+-----+ 

 

       1  2  3   4   5 

-----+-----------------+ 

ALOW | 1  4  6   9  11 | 

AHGH | 3  5  8  10  11 | 

-----+-----------------+ 

Sparse Matrix Times Vector 
A dense matrix times vector operation b += A x can be expressed as a sequence of        ·         
SDOT operations (favoring row-wise storage) or a sequence of SAXPY operations           
(favoring column-wise storage). In 1-D vectorized form, the two variants look as follows. 
 
DO I = 1, N                              DO J = 1, N 

  B(I) += SDOT(A(I,1:N), X(1:N))           B(1:N) += SAXPY(A(1:N,J), X(J)) 

ENDDO                                    ENDDO 

 

When converted to sparse code, first a suitable sparse storage scheme must be             
selected that best facilitates skipping redundant operations on zero elements. Using the            
previously discussed row- and column-wise storage schemes for, respectively, the          
SDOT and SAXPY versions, gives rise to the following code, with a memory indirection              
in the vector load and/or store operations. 
 

DO I = 1, N                              DO J = 1, N 

  L = ALOW(I)                              L = ALOW(J) 

  H = AHGH(I)                              H = AHGH(J) 

  B(I) += SDOT(AVAL(L:H), X(AIDX(L:H)))    B(AIDX(L:H)) += SAXPY(AVAL(L:H), X(J)) 

ENDDO                                    ENDDO 

 

The code reveals two typical complications in sparse applications that are relevant in             
the context of this case study. 
 

1. All vectors are variable-length. For example, for the first row, the index array and,              
thus, the value array and other arrays are accessed with vector length VL=3,             
whereas the second row gives rise to VL=2. The last row even has VL=1. 

2. Gather (indirect load) and scatter (indirect store) are needed to perform the            
computations in a sparse manner. Such irregular memory accesses exhibit poor           
spatial locality and require special instructions in order to exploit SIMD           
effectively. 



 

 
Complication (1) can be dealt with using a combination of vector.transfer_read/write and            
masking, or sometimes by choosing a more vector-friendly storage scheme (see next            
section). Complication (2) has been resolved by introducing vector.gather/scatter         
operations to the Vector dialect (perhaps even just vector.transfer_read/write would          
have sufficed here, but the explicit gather and scatter make progressive lowering to             
efficient instructions in the hardware ISA more transparent). 

Sparse Matrix Times Vector Benchmarks 
The benchmarks presented in this section focus on the performance of gather/scatter            
specifically by making a few simplifying changes to the general sparse storage scheme.             
First, the assumption is made that every row/column in the sparse matrix has a fixed,               
unified number of nonzero elements (or can be made so by moderately padding in a few                
zero elements), so that a fixed vector length can be used throughout the computation.              
Second, details related to the “pointer” arrays are abstracted away by assuming a             
fixed-length value and index vectors appear consecutively in memory. Note that with            
these vector-friendly changes, effectively the data structure has become an extended           
column/row scheme (also called ITPACK scheme or jagged diagonal storage). 
 
The extended column scheme (used for SDOT, which favors row-wise access) for the             
sparse example matrix is illustrated below (empty positions are padded with arbitrary            
zeros from the corresponding row, but left open here for illustration purposes). 
 
      AVAL                AIDX 

    +-------------+    +-------+ 

    | a11 a13 a14 |    | 1 3 4 | 

    | a22 a25     |    | 2 5   | 

    | a31 a33 a34 |    | 1 3 4 | 

    | a41 a44     |    | 1 4   | 

    | a55         |    | 5     | 

    +-------------+    +-------+ 

 

The extended row scheme (used for SAXPY, which favors column-wise access) is            
similar. Yes another jagged diagonal variant will be used for a second version of SAXPY               
as well. Test input matrices range in size from 16x16 to 512x512 with a relatively high                
sparsity of 50% (so that the number of nonzeros per row ranges from 8 to 256, and the                  
total number of nonzeros in the matrix ranges from 128 to 131072). 



 

Sparse SDOT Version (Extended Scheme) 
The 16x16 benchmark based on the sequence of SDOT operations solely expressed in             
the Vector dialect looks as follows. Benchmarks for the other shapes look similar, with              
the corresponding sizes hardcoded in (as allowed under our simplifying assumptions;           
production code will have to deal with a lot more details). 
 

  func @spmv16_sdot(%AVAL: memref<16xvector<8xf32>>, 

                    %AIDX: memref<16xvector<8xi32>>, 

                    %X: memref<?xf32>, %B: memref<?xf32>) { 

    %c0 = constant 0  : index 

    %c1 = constant 1  : index 

    %cn = constant 16 : index 

    %mask = vector.constant_mask [8] : vector<8xi1> 

    scf.for %i = %c0 to %cn step %c1 { 

      %aval = load %AVAL[%i] : memref<16xvector<8xf32>> 

      %aidx = load %AIDX[%i] : memref<16xvector<8xi32>> 

      %0 = vector.gather %X, %aidx, %mask 

        : (memref<?xf32>, vector<8xi32>, vector<8xi1>) -> vector<8xf32> 

      %b = load %B[%i] : memref<?xf32> 

      %1 = vector.contract 

         #dot_trait %aval, %0, %b : vector<8xf32>, vector<8xf32> into f32 

      store %1, %B[%i] : memref<?xf32> 

    } 

    return 

  } 

 

The AVX512 for the loop-body above looks as follows (under          
reassociate-fp-reductions). 
 

        vmovaps ymm0, ymmword ptr [rdx] 

        kxnorw  k1, k0, k0 

        vgatherdps ymm1 {k1}, ymmword ptr [rcx + 4*ymm0] 

        mov     rax, qword ptr [rsp + 88] 

        vmulps  ymm0, ymm1, ymmword ptr [rsi] 

        vextractf128 xmm1, ymm0, 1 

        vaddps    xmm0, xmm0, xmm1 

        vpermilpd xmm1, xmm0, 1           # xmm1 = xmm0[1,0] 

        vaddps    xmm0, xmm0, xmm1 

        vmovshdup xmm1, xmm0              # xmm1 = xmm0[1,1,3,3] 

        vaddss    xmm0, xmm0, xmm1 

        vaddss    xmm0, xmm0, dword ptr [rax] 

        vmovss    dword ptr [rax], xmm0 



 

Sparse SAXPY Version (Extended Scheme) 
The 16x16 benchmark based on the sequence of SAXPY operations looks as follows.             
Benchmarks for other shapes look similar, with the corresponding sizes hardcoded in.            
Note that the indirection into the vector b update requires a gather and a scatter. 
 

  func @spmv16_saxpy(%AVAL: memref<16xvector<8xf32>>, 

                     %AIDX: memref<16xvector<8xi32>>, 

                     %X: memref<?xf32>, %B: memref<?xf32>) { 

    %c0 = constant 0  : index 

    %c1 = constant 1  : index 

    %cn = constant 16 : index 

    %mask = vector.constant_mask [8] : vector<8xi1> 
    scf.for %j = %c0 to %cn step %c1 { 

      %aval = load %AVAL[%j] : memref<16xvector<8xf32>> 

      %x = load %X[%j] : memref<?xf32> 

      %xv = vector.broadcast %x : f32 to vector<8xf32> 

      %aidx = load %AIDX[%j] : memref<16xvector<8xi32>> 

      %0 = vector.gather %B, %aidx, %mask 

         : (memref<?xf32>, vector<8xi32>, vector<8xi1>) -> vector<8xf32> 

      %1 = vector.fma %aval, %xv, %0 : vector<8xf32> 

      vector.scatter %B, %aidx, %mask, %1 

         : vector<8xi32>, vector<8xi1>, vector<8xf32> into memref<?xf32> 

    } 

    return 

  } 

 

This yields the following AVX512 instructions. 
 
        vmovaps ymm0, ymmword ptr [rsi] 

        vmovaps ymm1, ymmword ptr [rcx] 

        kxnorw  k1, k0, k0 

        vgatherdps ymm2 {k1}, ymmword ptr [rax + 4*ymm1] 

        mov     rdx, qword ptr [rsp + 48] 

        vfmadd231ps ymm2, ymm0, dword ptr [rdx]{1to8} 

        kxnorw  k1, k0, k0 

        vscatterdps ymmword ptr [rax + 4*ymm1] {k1}, ymm2 

Sparse SAXPY Version (Jagged Diagonal Variant) 
The performance of a straightforward SAXPY version suffers from the requirement of            
performing a gather and a scatter for each vector operation. With jagged diagonal             
storage this can be very easily remedied by performing vector operations along the             
jagged column diagonals in the sparse matrix. 



 

This jagged diagonal variant gives rise to the following 16x16 benchmark for SAXPY. 
 
  func @spmv16(%AVAL: memref<8xvector<16xf32>>, 

               %AIDX: memref<8xvector<16xi32>>, 

               %X: memref<?xf32>, 

               %B: memref<1xvector<16xf32>>) { 

    %c0 = constant 0 : index 

    %c1 = constant 1 : index 

    %cn = constant 8 : index 

    %mask = vector.constant_mask [16] : vector<16xi1> 

    %b = load %B[%c0] : memref<1xvector<16xf32>> 

    %b_out = scf.for %k = %c0 to %cn step %c1 

          iter_args(%b_iter = %b) -> (vector<16xf32>) { 

      %aval = load %AVAL[%k] : memref<8xvector<16xf32>> 

      %aidx = load %AIDX[%k] : memref<8xvector<16xi32>> 

      %0 = vector.gather %X, %aidx, %mask 

         : (memref<?xf32>, vector<16xi32>, vector<16xi1>) -> vector<16xf32> 

      %b_new = vector.fma %aval, %0, %b_iter 

         : vector<16xf32> 

      scf.yield %b_new : vector<16xf32> 

    } 

    store %b_out, %B[%c0] : memref<1xvector<16xf32>> 

    return 

  } 

  
This yields very compact AVX512 code within the loop-body, keeping the vector b fully              
contained within SIMD registers during the SAXPY accumulation before storing it back.            
Part of the fully unrolled loop is shown below. 
 
        vmovaps zmm0, zmmword ptr [rcx] 

        kxnorw  k1, k0, k0 

        vgatherdps  zmm4 {k1}, zmmword ptr [rax + 4*zmm0] 

        vfmadd213ps zmm4, zmm3, zmmword ptr [rdx] 

Performance Graphs 
Before reporting performance numbers, please keep in mind that because sparse           
computations use elaborate constructs to reduce the total execution time by skipping            
operations on zeros, the observed GFlops rates are typically much lower than the peak              
rates observed for dense computations. Obtaining lower execution time is the ultimate            
objective, but the GFlops rate still gives a measure of how fast that goal is achieved. 
 
 



 

Figure 1 shows the performance in GFlops for all versions with JIT running on an               
AVX2-enabled Intel Xeon E5-2690 2.6GHz desktop (see sdot/saxpy/saxpy-fma). The         
jagged diagonal SAXPY version performs best up to vector length VL=64 (see            
saxpy-jagged). It is still an open question how close to sparse peak performance this is.               
But since libraries like MKL and Eigen perform between 2 to 4 GFlops using a general                
CRS format on the same sparse matrices, results are promising! Figure 2 shows the              
performance on an AVX512-enabled Intel Xeon W-2135 3.7GHz desktop. 

 
 Figure 1. SpMxV Performance (AVX2) 

 
Figure 2. SpMxV Performance (AVX512) 



 

Next Steps 
This case-study was just a first small step in exploring proper operational support for              
sparse computations in the Vector dialect. Basic gather/scatter support was this first            
step, combined with some performance analysis of sparse matrix times vector           
multiplication using jagged diagonal storage. Follow up case-studies should focus on           
other memory-related operations and how they can benefit sparse computations. 
 
More serious follow-up research for sparse computation support in MLIR is planned            
next as well. The impact of different storage schemes, runtime values, sparsity            
percentages, and nonzero structures need to be studied in more detail. Investigation of             
other kernels, like sparse matrix times matrix, is planned too, since such kernels may              
allow amortizing the gather/scatter overhead better as well. Lastly, in the context of ML,              
we started exploring annotating dense tensors as sparse somehow, possibly by specifically            
indicating a particular sparse storage scheme or leaving that open for the compiler to              
decide, and let the compiler subsequently generate sparse code that operates on these             
tensors automatically. This approach has many advantages. Users can focus on the            
algorithm rather than the data structures, one dense program can be mapped to several              
different sparse storage formats, possibly tailored for particular nonzero structures. 


